
Tutorial Completo – Curva ABC de Fornecedores no
Power BI

1. Visão geral
Neste tutorial vamos montar, passo a passo, uma Curva ABC de Fornecedores no Power BI. O objetivo é
classificar cada fornecedor como A, B ou C com base no valor total comprado no período e no percentual
acumulado. Vamos construir as medidas de valor, participação, soma acumulada e, por fim, a
classificação ABC.

2. Estrutura de dados necessária
Para seguir o tutorial, considere que você possui:

• Uma tabela de fatos (por exemplo, Fato_Fornecedores) com as compras por mês e por fornecedor.
• Uma dimensão de fornecedores (Fornecedores), com pelo menos a coluna Fornecedor.

Na tabela de fatos, vamos assumir que existe a coluna ValorTotalMes, que representa o valor total
comprado daquele fornecedor em determinado mês.

3. Criando as medidas básicas

3.1 Medida Total Compras
Total Compras =
SUM ( Fato_Fornecedores[ValorTotalMes] )

3.2 Medida Total Compras Geral
Total Compras Geral =
CALCULATE (
[Total Compras],
ALL ( Fornecedores )
)

3.3 Medida Valor Fornecedor
Valor Fornecedor = [Total Compras]

4. Percentual de participação do fornecedor
% Participação =
DIVIDE ( [Valor Fornecedor], [Total Compras Geral] )

Essa medida retorna um valor entre 0 e 1 (que você pode formatar como porcentagem). Por exemplo, se
um fornecedor tem R$ 100.000 em compras e o total geral é R$ 1.000.000, a % Participação será 0,10
(10%).



5. Soma acumulada por fornecedor (ponto mais
importante)
Para a Curva ABC, o segredo está em calcular o valor acumulado respeitando a ordem decrescente de
valor dos fornecedores.

5.1 Medida Valor Fornecedor Acumulado
Valor Fornecedor Acumulado =
VAR ValorAtual = [Valor Fornecedor]

RETURN
CALCULATE (
[Valor Fornecedor],
FILTER (
ALL ( Fornecedores[Fornecedor] ),
[Valor Fornecedor] >= ValorAtual
)
)

5.2 Entendendo a lógica com um exemplo numérico
Imagine a lista de fornecedores com o valor total de compras já ordenado em ordem decrescente (do
maior para o menor):

• Fornecedor A – R$ 400.000
• Fornecedor B – R$ 250.000
• Fornecedor C – R$ 200.000
• Fornecedor D – R$ 100.000
• Fornecedor E – R$ 50.000

Total geral = R$ 1.000.000

Quando o Power BI está avaliando a medida para o Fornecedor A:

• ValorAtual = 400.000
• FILTER pega todos os fornecedores com valor maior ou igual a 400.000 (apenas A).
• Soma = 400.000 → Valor Fornecedor Acumulado (A) = 400.000.

Para o Fornecedor B:

• ValorAtual = 250.000
• Filtro: A e B (valores ≥ 250.000).
• Soma = 400.000 + 250.000 = 650.000 → Valor Fornecedor Acumulado (B) = 650.000.

Para o Fornecedor C:

• ValorAtual = 200.000
• Filtro: A, B e C (valores ≥ 200.000).
• Soma = 400.000 + 250.000 + 200.000 = 850.000.

O padrão se repete para D e E, até atingir o total geral de 1.000.000. ALL(Fornecedores[Fornecedor])
remove o filtro de linha, permitindo que o CALCULATE veja todos os fornecedores ao mesmo tempo, e a



condição [Valor Fornecedor] >= ValorAtual define quem entra na soma.

6. Percentual acumulado do fornecedor
% Acumulado Fornecedor =
DIVIDE ( [Valor Fornecedor Acumulado], [Total Compras Geral] )

Com o total geral de R$ 1.000.000, teríamos por exemplo:
• A – 400.000 / 1.000.000 = 40%
• B – 650.000 / 1.000.000 = 65%
• C – 850.000 / 1.000.000 = 85%
• D – 950.000 / 1.000.000 = 95%
• E – 1.000.000 / 1.000.000 = 100%

7. Classificando a Curva ABC
Classe ABC =
SWITCH (
TRUE(),
[% Acumulado Fornecedor] <= 0.80, "A",
[% Acumulado Fornecedor] <= 0.95, "B",
"C"
)

• Classe A: fornecedores que, acumulados, chegam até 80% do total.
• Classe B: fornecedores entre 80% e 95%.
• Classe C: fornecedores que completam de 95% a 100%.

8. Montando o visual da Curva ABC
1. Crie uma tabela ou matriz no relatório e coloque Fornecedores[Fornecedor].
2. Adicione as medidas: [Valor Fornecedor], [% Participação], [Valor Fornecedor Acumulado], [%
Acumulado Fornecedor] e [Classe ABC].
3. Ordene a visualização por [Valor Fornecedor] em ordem decrescente.
4. Formate os percentuais e ajuste os cabeçalhos para facilitar a leitura.
5. Use segmentações (filtros de período, região, etc.) para analisar cenários diferentes.

9. Conclusão
Com esse conjunto de medidas você tem uma Curva ABC completa no Power BI, permitindo enxergar
rapidamente quais fornecedores respondem pela maior parte das compras e onde concentrar esforços de
negociação e acompanhamento.


